

## Filter 1 (Einzelne Items ausblenden)

### Einleitung

- In dieser Folge blenden wir einzelne Items in unserer Pivot Tabelle aus.
- Zum besseren Verständnis dieser Folge sind in der "Anlage 1 – Begrifflichkeiten der Pivot Tabelle" die Begriffe *PivotTable*, *PivotField* und *PivotItem* nochmals definiert.

### Ausgangssituation

- Tabellenblatt "QUELLTABELLE"

	A	B	C	D	E	F	G	H	I	J	K
1	Datum	Produkt	Betrag	Land	Geschlecht	Zahlungsart	Feedback				
2	01.07.2020	P1 Weinflasche	46	Deutschland	M	PayPal	5		Pivot Tabelle erstellen		
3	01.07.2020	P2 Weinglas	31,99	Drittland	W	PayPal	4				
4	01.07.2020	P3 Korkenzieher	14,25	Deutschland	W	PayPal	3				
5	01.07.2020	P1 Weinflasche	36	EU	W	Rechnung	4				
6	01.07.2020	P2 Weinglas	31,99	Deutschland	M	PayPal	3				
7	01.07.2020	P1 Weinflasche	46	Drittland	M	Rechnung	4				
8	01.07.2020	P1 Weinflasche	46	EU	W	Rechnung	5				
9	01.07.2020	P2 Weinglas	31,99	EU	W	PayPal	4				
10	01.07.2020	P1 Weinflasche	46	Deutschland	M	Rechnung	3				
11	01.07.2020	P3 Korkenzieher	12,5	Deutschland	W	PayPal	2				
12	01.07.2020	P1 Weinflasche	46	Deutschland	W	PayPal	4				
13	01.07.2020	P1 Weinflasche	46	EU	W	PayPal	5				
14	02.07.2020	P2 Weinglas	25	Deutschland	M	PayPal	4				
15	02.07.2020	P3 Korkenzieher	14,25	Drittland	M	PayPal	4				
16	02.07.2020	P1 Weinflasche	46	Deutschland	M	Nachname	3				
17	02.07.2020	P2 Weinglas	31,99	Deutschland	M	PayPal	1				
18	02.07.2020	P3 Korkenzieher	14,25	Drittland	W	PayPal	2				
19	02.07.2020	P1 Weinflasche	36	Deutschland	W	PayPal	5				
20	02.07.2020	P2 Weinglas	31,99	EU	W	Nachname	2				
21	02.07.2020	P1 Weinflasche	46	Deutschland	W	PayPal	3				
22	03.07.2020	P1 Weinflasche	46	Drittland	W	Rechnung	4				

### Einzelne Items ausblenden

#### Unsere Pivot Tabelle in einem anderen Modul ansprechen

- Für die folgenden Beispiele dieses Videokurses betrachten wir zunächst, wie wir unsere Pivot Tabellen in einer neuen Sub Prozeduren ansprechen können.

Modul1
Sub PivotTabelleAnsprechen()  <b>'Die PivotTabelle in unserem Tabellenblatt ansprechen (mit Nummer)</b> MsgBox ActiveSheet.PivotTables(1).Name  <b>'Die PivotTabelle in unserem Tabellenblatt ansprechen (mit Namen)</b> MsgBox ActiveSheet.PivotTables("Meine PivotTabelle").Name  End Sub

## Filter 1 (Einzelne Items ausblenden)

### Beispiel 1: Einzelne Items ausblenden

- Wir blenden die beiden Items „P2 Weinglas“ und „P3 Korkenzieher“ aus

```
Modul1
Sub ItemsFiltern()

    'Die Items P2 Weinglas und P3 Korkenzieher ausblenden
    With ActiveSheet.PivotTables(1).PivotFields("Produkt")
        .PivotItems("P3 Korkenzieher").Visible = False
        .PivotItems("P2 Weinglas").Visible = False
    End With

End Sub
```

### Beispiel 2: Einzelne Items einblenden

- Anschließend blenden wir das Item „P2 Weinglas“ wieder ein

```
Modul1
Sub ItemsFiltern()

    'Das Item "Korkenzieher" ausblenden / "Weinglas" einblenden
    With ActiveSheet.PivotTables(1).PivotFields("Produkt")
        .PivotItems("P3 Korkenzieher").Visible = False
        .PivotItems("P2 Weinglas").Visible = True
    End With

End Sub
```

### Variablen definieren für PivotTable, PivotField, PivotItem

- In diesem Programm erstellen wir die drei Variablen *pt*, *pf* und *pi*, welche wir in den folgenden Videos immer mal wieder nutzen werden. Zudem betrachten wir den Befehl "EnableMultiplePageItems". Dessen Wert ist Standardmäßig *True* und ermöglicht es uns, mehrere Items auszuwählen.

```
Modul1
Sub ItemsFiltern()

    'Variablen definieren
    Dim pt As PivotTable    'Variable für die Pivot Tabelle
    Dim pf As PivotField    'Variable für die Tabellenüberschriften
    Dim pi As PivotItem     'Variable für die Tabelleneinträge

    'Werte zuweisen
    Set pt = ActiveSheet.PivotTables(1)
    Set pf = pt.PivotFields("Produkt")
    Set pi = pf.PivotItems("P3 Korkenzieher")

    'Erlauben, dass mehrere Items ausgewählt werden dürfen
    pf.EnableMultiplePageItems = True

    'Item ausblenden
    pi.Visible = False

End Sub
```



## Filter 1 (Einzelne Items ausblenden)

### Alle Items einer Tabellenüberschrift betrachten - Schleife einbauen

- Diesmal weisen wir der Variable pi keinen festen Wert zu. Wir gehen mit der For Each Schleife alle Items der Produkte durch und lassen uns den Namen der Produkte in einer MsgBox ausgeben.

```
Modul1

Sub ItemsFiltern()

    'Variablen definieren
    Dim pt As PivotTable 'Variable für die Pivot Tabelle
    Dim pf As PivotField 'Variable für die Tabellenüberschriften
    Dim pi As PivotItem 'Variable für die Tabelleneinträge

    'Werte zuweisen
    Set pt = ActiveSheet.PivotTables(1)
    Set pf = pt.PivotFields("Produkt")

    'Alle Items (Produkte) anzeigen
    For Each pi In pf.PivotItems
        MsgBox pi.Name
        pi.Visible = True
    Next pi

End Sub
```

#### ▶ Videoempfehlung

„EXCEL VBA Die wichtigsten Befehle #19 - For Each Schleifen“

---

In diesem Video wird die For Each Schleife anhand mehrere Beispiele vorgestellt.

## Filter 1 (Einzelne Items ausblenden)

### Beispiel 3: Alle Items einer Tabellenüberschrift betrachten

- Mit der For Each Schleife gehen wir alle PivotItems durch, und prüfen, ob der Name "P1 Weinflasche" oder "P2 Weinglas" lautet.

```
Modul1

Sub ItemsFiltern()

    'Variablen definieren
    Dim pt As PivotTable 'Variable für die Pivot Tabelle
    Dim pf As PivotField 'Variable für die Tabellenüberschriften
    Dim pi As PivotItem 'Variable für die Tabelleneinträge

    'Werte zuweisen
    Set pt = ActiveSheet.PivotTables(1)
    Set pf = pt.PivotFields("Produkt")

    'Alle Items anzeigen
    For Each pi In pf.PivotItems
        Select Case pi.Name
            Case "P1 Weinflasche", "P2 Weinglas"
                'Falls P1 oder P2: Item anzeigen
                pi.Visible = True
            Case Else
                'P3: Item ausblenden
                pi.Visible = False
        End Select
    Next pi
End Sub
```

#### ▶ Videoempfehlung

„EXCEL VBA Die wichtigsten Befehle #17 - Select Case Verzweigungen“

---

In diesem Video werden die Select Case Verzweigungen anhand mehrere Beispiele vorgestellt.

### Beispiel 4: Alle Filtereinstellungen löschen

- Über diese Prozedur können wir alle Filtereinstellungen rückgängig machen.

```
Modul1

Sub FilterLoeschen()

    'Filter löschen aus „Produkt“
    ActiveSheet.PivotTables(1).PivotFields("Produkt").ClearAllFilters

    'Alle Filter löschen
    ActiveSheet.PivotTables(1).ClearAllFilters

End Sub
```

#### Hinweis

Es kann vorkommen, dass gelöschte PivotFields weiterhin auftauchen. Hierzu befindet sich in der Anlage 2 eine Anleitung, wie Du dies über Excel und Excel VBA verhindern kannst.